

5.4 Excursion: Logical Framework LF

The logical framework LF [Harper et al.(1993)Harper, Honsell, and Plotkin] is system for defining logics and formal systems based on the dependently-typed lambda-calculus. It is also sometimes referred to as $\lambda\Pi$ -calculus. From a proof-theoretic point of view, it is equivalent to the fragment of first-order logic consisting of universal quantification, implication, and base predicates. In fact, even universal quantification and implication are collapsed when we interpret it from a type-theoretic perspective! Hence, while the simply-typed lambda-calculus has base types (proposition) and function space (implications), the $\lambda\Pi$ -calculus has base type families (predicates) and dependent function space (universal quantification).

While very compact and small, it is an expressive framework that allows us to elegantly encode formal systems and logics. In particular, it is rich enough to encode and represent our natural deduction rules for first-order logic inside it! In fact there is a one-to-one correspondance between what proofs are described “on-paper” in our proof system and the objects described using the encoding of the proof system in LF. This is an important and often hard to achieve property.

5.4.1 Grammar and Typing

The logical framework LF allows us to define new types. Just as types classify terms, we can classify types using kinds. Let us start with the terms in LF. To characterize only terms in normal form, we split them into normal terms and neutral terms. Normal terms are either lambda-abstraction or a neutral term R . Neutral terms are either variable, constant, or an application of a neutral term to a normal term.

Kind	K	$::=$	$\text{type} \mid \Pi x:A.K$
Type	A, B	$::=$	$a \ M_1 \dots M_n \mid \Pi x:A.B$
Normal Term	M, N	$::=$	$\lambda x.M \mid R$
Neutral Term	R	$::=$	$x \mid c \mid R \ M$
Signature	Σ	$::=$	$\cdot \mid \Sigma, a : K \mid \Sigma c : A$

Types are declared to have a kind. The simple types have kind type. For example, we might declare a new type `nat` by stating in a signature that `nat : type`. Similarly, we can declare types that depend on arguments which are called type families. Such type families correspond logically to predicates. For example, we might want to declare that the type family (predicate) `even` takes in a natural number. This can be done by defining in the signature: `even : $\Pi n:\text{nat}.\text{type}$` . The type family (predicate) `lt` that relates and compares two natural numbers, can be defined as: `lt : $\Pi n:\text{nat}.\Pi n:\text{nat}.\text{type}$` .

We simply write $A \rightarrow K$ instead of $\Pi x:A.K$, if x does not occur in the free in K . Hence, the previous three definitions can simply be stated as:

$$\begin{aligned} \text{nat} & : \text{type} \\ \text{even} & : \text{nat} \rightarrow \text{type} \\ \text{lt} & : \text{nat} \rightarrow \text{nat} \rightarrow \text{type} \end{aligned}$$

Types correspond logically to propositions. They are formed using atomic types $a M_1 \dots M_n$ where a stands for a type constant (or type family or logically a predicate) and $M_1 \dots M_n$ are the arguments. How many arguments a type constant accepts is determined by its kind. For example, the type constant nat is declared simply of kind type and takes in no arguments; the type constant even takes in a natural number as an argument; the type constant lt takes in two natural numbers.

We can declare in the signature term constants together with their type. Here are a few examples:

$$\begin{aligned} z & : \text{nat}. \\ s & : \text{nat} \rightarrow \text{nat}. \end{aligned}$$

We again write simply $A \rightarrow B$ instead of $\Pi x:A.B$ if x does not occur in B . Hence, the dependent function space, called Π -type, degenerates to the simply typed function space $A \rightarrow B$.

$$\begin{aligned} \text{ev_z} & : \text{even } z. \\ \text{ev_s} & : \Pi N:\text{nat}. \text{even } N \rightarrow \text{even } (s(s \ N)). \end{aligned}$$

Let us show the typing rules. We write P or Q for atomic types.

$\boxed{\Gamma \vdash M \Leftarrow A}$ Normal Term M checks against type A .

$$\frac{\Gamma, x:A \vdash M \Leftarrow B}{\Gamma \vdash \lambda x.M \Leftarrow \Pi x:A.B} \text{abs} \quad \frac{\Gamma \vdash R \Rightarrow a M_1 \dots M_n}{\Gamma \vdash R \Leftarrow A M_1 \dots M_n} \text{coe}$$

$\boxed{\Gamma \vdash R \Rightarrow A}$ Neutral Term R synthesizes type A .

$$\frac{x:A \in \Gamma}{\Gamma \vdash x \Rightarrow A} \text{var} \quad \frac{c:A \in \Sigma}{\Gamma \vdash c \Rightarrow A} \text{const} \quad \frac{\Gamma \vdash R \Rightarrow \Pi x:A.B \quad \Gamma \vdash M \Leftarrow A}{\Gamma \vdash R M \Rightarrow [M/x]B} \text{app}$$

The typing rules should look very familiar, as they correspond to the annotated normal proofs for the fragment of first order logic that only considers universal quantifiers and implication, where we collapse the rules for implications and universal quantification.

We hence employ *hereditary substitution* operation which replaces x by M and eliminates all possible redexes as we go along. This operation goes back to [Watkins et al.(2002)Watkins]. We omit the precise definition of *hereditary substitution* here, but refer the reader to for example [Cave and Pientka(2012)] or similar work.

$$\text{ev_s } (s \ (s \ z)) \ (\text{ev_s } z \ \text{ev_z}) \ : \ \text{even } (s \ (s \ (s \ (s \ z))))$$

Using this derivation in the derivation below, we can justify :

As this example shows in detail, we can use the rules of the $\lambda\Pi$ -calculus to encode our own theories by defining new types and type families as well as constants inhabiting them.

5.4.2 Beluga: a proof and programming environment based on the logical framework LF

Beluga [Pientka and Dunfield(2010), Pientka and Cave(2015)] provides an implementation of the logical framework LF. The previous example of natural numbers and definition of even, can be encoded in Beluga's concrete syntax as follows:

```
LF nat: type =
| z : nat
| s : nat → nat;

LF even: nat → type =
| ev_z : even z
| ev_s : even N → even (s (s N));
```

The keyword **LF** states we are introducing an LF type (family) together with constants of that type. For convenience, we omitted the universal quantifier (written using Π in the previous section), but by convention all free variables in a given type are universally quantified at the outside implicitly. Beluga's reconstruction engine will infer the type of N and abstract over N at the outside. We hence elaborate the type $\text{even } N \rightarrow \text{even } (s (s N))$ to $\Pi N:\text{nat}. \text{even } N \rightarrow \text{even } (s (s N))$.

Beluga further has a type checker that verifies that a given term has a given type following the typing rules from the previous section. The type checker is clever enough that we can omit passing instantiations for N when we rely on the Π -elimination rule in our derivation. Instead of $\vdash \text{ev}_s z \text{ ev}_z \Leftarrow \text{even } (s (s z))$ we simply write the following concrete expression in Beluga:

```
rec d1 : [ ⊢ even (s (s z))] = [ ⊢ ev_s ev_z];
```

We use **rec** as a keyword of introducing constants that stand for proofs. Note that we omit writing z and do not pass it as an argument to ev_s explicitly; however, type reconstruction in Beluga will infer it. This makes derivations more readable. We refer to LF objects inside $[]$. The turnstyle \vdash is used to separate assumptions from the main statement we are trying to establish. In the given example, all derivations were closed and did not rely on hypothesis.

Let us look at a few more examples.

Encoding Addition We can encode addition as a type family relating three natural numbers as follows:

```
LF add: nat → nat → nat → type =
| a_z: add z N N
| a_s: add N M K → add (s N) M (s K);
```

We again omit writing Π -quantification over N , M , and K explicitly, and let Beluga's reconstruction engine infer the type of the free variables and abstract over them implicitly. Then we can state the proof of the fact that adding $s \ (s \ z)$ to a variable N results in $s \ (s \ N)$.

```
rec a1 : [  $\vdash$  add (s (s z)) N (s (s N)) ] = [  $\vdash$  a_s (a_s a_z) ];
```

Beluga enforces the use of capital letters for universally quantified variables. For example, the previous statement is for all N that are natural numbers.

Encoding Vectors We can encode boolean vectors as lists that are indexed by their length as follows using a type family `vec` that takes in `nat` as an argument. We then construct vectors either with the empty vector described by the constant `e` or by the constant `snoc` that takes in a vector of length N and a `bool` and constructs a vector of length $s \ N$.

```
LF bool : type =
| tt: bool
| ff: bool;

LF vec: nat  $\rightarrow$  type =
| e: vec z
| snoc : vec N  $\rightarrow$  bool  $\rightarrow$  vec (s N);
```

We can then for example possible vectors of length $(s \ (s \ z))$.

```
rec v0 : [  $\vdash$  vec (s (s z)) ] = [  $\vdash$  snoc (snoc e tt) ff ];
rec v1 : [  $\vdash$  vec (s (s z)) ] = [  $\vdash$  snoc (snoc e ff) ff ];
rec v2 : [  $\vdash$  vec (s (s z)) ] = [  $\vdash$  snoc (snoc e ff) tt ];
rec v3 : [  $\vdash$  vec (s (s z)) ] = [  $\vdash$  snoc (snoc e tt) tt ];
```

Encoding Natural Deduction We now encode the natural deduction rules for propositional logic in LF. This will allow us to encode and check natural deduction proofs using the simple typing rules of LF. We first define a type `o` describing logical propositions we want to represent.

```
LF o : type =
| imp : o  $\rightarrow$  o  $\rightarrow$  o
| & : o  $\rightarrow$  o  $\rightarrow$  o
| top : o;
```

Let's revisit our introduction and elimination rules for these propositions.

$$\begin{array}{c}
\boxed{A \text{ true}} \quad A \text{ is true} \\
\\
\frac{}{\top \text{ true}} \top I \quad \frac{A \supset B \text{ true} \quad A \text{ true}}{B \text{ true}} \supset E \quad \frac{\frac{\frac{}{A \text{ true}} u \quad \vdots \quad B \text{ true}}{A \supset B \text{ true}} \supset I^u}{} \\
\\
\frac{A \text{ true} \quad B \text{ true}}{A \wedge B \text{ true}} \wedge I \quad \frac{A \wedge B \text{ true}}{A \text{ true}} \wedge E_l \quad \frac{A \wedge B \text{ true}}{B \text{ true}} \wedge E_r
\end{array}$$

We represent the judgment $A \text{ true}$ using the type family nd . Each inference rule turns into a constant of the type family nd .

```

LF nd: o → type.
| andI : nd A → nd B → nd (& A B)
| andEl : nd (& A B) → nd A
| andEr : nd (& A B) → nd B
| impI : (nd A → nd B) → nd (imp A B)
| impE : nd (imp A B) → nd A → nd B
| topI : nd top;

```

Some of these constant definitions are straightforward. For example, the constant `andI` directly encodes the inference rule $\wedge I$. Similarly, the constant `andEl` and `andEr` correspond directly to the inference rules $\wedge E_l$ and $\wedge E_r$. The constant `topI` corresponds directly to the rule $\lambda_{\text{top}} I$.

The most interesting is the encoding of the hypothetical sub-derivation in the rule $\supset I$. How shall we encode the hypothetical derivation “given the hypothesis $A \text{ true}$ we must derive a proof of $B \text{ true}$ ”? - Here is the trick: LF has function spaces. We map the hypothetical derivation “given the hypothesis $A \text{ true}$ we must derive a proof of $B \text{ true}$ ” to the function $\text{nd } A \rightarrow \text{nd } B$. This form of encoding is called higher-order abstract syntax as our abstract syntax trees may contain functions which we use to model the scope of hypothesis in hypothetical derivations. Functions in Beluga are written as $\lambda u. M$ where M is the body of the function and we are abstracting over the variable u .

Let’s look at a few examples.

```

rec p0 : [ ⊢ nd (imp (& A B) A) ] =
  [ ⊢ impI (λu. andEl u) ] ;
rec p1 : [ ⊢ nd (imp (& A B) (& B A)) ] =
  [ ⊢ impI λu. andI (andEr u) (andEl u) ];
rec p2 : [ ⊢ nd (imp (& (imp A B) (imp B C)) (imp A C)) ] =
  [ ⊢ impI λu. (impI λv. impE (andEr u) (impE (andEl u) v )) ];

```

The astute reader will observe that unlike Tutch where we write in each line the formula that can be derived from the previous lines, we here write only the justifications (i.e. `impI` or `andEl`), but not the formula itself. Beluga's type checker will make sure that the given justifications indeed constitute a valid proof. The scope of an assumption is introduced using an LF-function written as $\lambda u.M$.

It is instructive to give different names to constants and type family and given them more suggestive names. Let's do the following replacement

Original name	Replacement
<code>nd</code>	<code>tm</code>
<code>andI</code>	<code>pair</code>
<code>andEl</code>	<code>fst</code>
<code>andEr</code>	<code>snd</code>
<code>impI</code>	<code>lam</code>
<code>impE</code>	<code>app</code>
<code>topI</code>	<code>unit</code>

This results in the equivalent definitions that might be more familiar to read:

```

LF tm : o → type =
| pair : tm A → tm B → tm (& A B)
| fst : tm (& A B) → tm A
| snd : tm (& A B) → tm B
| lam : (tm A → tm B) → tm (imp A B)
| app : tm (imp A B) → tm A → tm B
| unit : tm top;

rec t0 : [ ⊢ tm (imp (& A B) A) ] =
  [ ⊢ lam (λu. fst u) ] ;
rec t1 : [ ⊢ tm (imp (& A B) (& B A)) ] =
  [ ⊢ lam λu. pair (snd u) (fst u) ];
rec t2 : [ ⊢ tm (imp (& (imp A B) (imp B C)) (imp A C)) ] =
  [ ⊢ lam λu. (lam λv. app (snd u) (app (fst u) v )) ];

```

Here we view the definition of natural deduction proofs as a definition of well-typed terms.

As this discussion illustrates, the logical framework LF acts like a *universal* proof checker. We encode our logic, for example our rules about even numbers, addition, or our natural deduction rules, and then can encode derivations as LF objects and use type checking to verify that a given derivation is valid.

Chapter 6

Induction

So far, we have seen first-order logic together with its corresponding proof-terms. First-order logic corresponds to the dependently typed lambda-calculus. However, if we are to write meaningful programs we need two more ingredients: 1) we need to reason about specific domains such as natural numbers, lists, etc 2) we need to be able to write recursive programs about elements in a given domain. Proof-theoretically, we would like to add the power of induction which as it turns out corresponds to being able to write total well-founded recursive programs.

6.1 Domain: natural numbers

First-order logic is independent of a given domain and the reasoning principles we defined hold for any domain. Nevertheless, it is useful to consider specific domains. There are several approaches to incorporating domain types or index types into our language: One is to add a general definition mechanism for *recursive types* or *inductive types*. We do not consider this option here, but we return to this idea later. Another one is to use the constructs we already have to define data. This was Church's original approach; he encoded numerals, booleans as well as operations such as addition, multiplication, if-statements, etc. as lambda-terms using a *Church encoding*. We will not discuss this idea in these notes. A third approach is to specify each type directly by giving rules defining how to construct elements of a given type (introduction rule) and how to reason with elements of a given type (elimination rule). This is the approach we will be pursuing here.

We begin by defining concretely the judgement

$t : \tau$ Term t has type τ

which we have left more or less abstract for now for concrete instances of τ .

6.1.1 Defining for natural numbers

We define elements belonging to natural numbers via two constructors z and succ . They allow us to introduce natural numbers. We can view these two rules as introduction rules for natural numbers.

$$\frac{}{z : \text{nat}} \text{natI}_z \quad \frac{t : \text{nat}}{\text{succ } t : \text{nat}} \text{natI}_{\text{succ}}$$

6.1.2 Reasoning about natural numbers

To prove inductively a property $A(t)$ true, we establish three things:

1. t is a natural number and hence we know how to split it into different cases.
2. *Base case:* $A(z)$ true
Establish the given property for the number z
3. *Step case:* For any $n : \text{nat}$, assume $A(n)$ true (I.H) and prove $A(\text{succ } n)$ true.
We assume the property holds for smaller numbers, i.e. for n , and we establish the property for $\text{succ } n$.

More formally, the inference rule capturing this idea is given below:

$$\frac{\begin{array}{c} \frac{}{n : \text{nat}} \quad \frac{}{A(n) \text{ true}} \text{i.h} \\ \vdots \\ t : \text{nat} \quad A(z) \text{ true} \quad A(\text{succ } n) \text{ true} \end{array}}{A(t) \text{ true}} \text{natE}^{n,\text{i.h}}$$

Restating the rule using explicit contexts to localize all assumptions:

$$\frac{\Gamma \vdash t : \text{nat} \quad \Gamma \vdash A(z) \text{ true} \quad \Gamma, n:\text{nat}, \text{ih}:A(n) \text{ true} \vdash A(\text{succ } n) \text{ true}}{\Gamma \vdash A(t) \text{ true}} \text{natE}^{n,\text{i.h}}$$

Let us prove a simple property inductively, to see how we use the rule.

Note the rule $\text{natE}^{n,\text{i.h}}$ has implicitly a generalization built-in. If we want to establish a property $A(42)$ true, we may choose to prove it more generally for any number

proving that $A(z)$ true (i.e. the property A holds for z) and proving $A(\text{succ } n)$ true (i.e. the property A holds for $\text{succ } n$) assuming the property A holds for n . Since we now have proven that A holds for all natural numbers, it also must hold for 42.

Let's look at an example; we first encode definitions which will form our signature \mathcal{S} .

$$\begin{aligned} \text{le_z} & : \forall x:\text{nat}. z \leq x \\ \text{le_suc} & : \forall x:\text{nat}. \forall y:\text{nat}. x \leq y \supset \text{succ } x \leq \text{succ } y \end{aligned}$$

Then we would like to prove: $\mathcal{S} \vdash \forall x:\text{nat}. x \leq x$ true. For better readability, we omit true in the derivations below.

$$\frac{\frac{\frac{}{\mathcal{S}, a:\text{nat} \vdash a : \text{nat}}}{} \quad \frac{}{\mathcal{S}, a:\text{nat} \vdash z \leq z} \mathcal{D}_z \quad \frac{}{\mathcal{S}, a:\text{nat}, n:\text{nat}, \text{ih}:n \leq n \vdash \text{succ } n \leq \text{succ } n} \mathcal{D}_{\text{succ}}}{\mathcal{S}, a:\text{nat} \vdash a \leq a} \text{natE}^{n,\text{ih}} \quad \frac{}{\mathcal{S} \vdash \forall x:\text{nat}. x \leq x} \forall \text{I}^a$$

We consider the base case \mathcal{D}_z and the step case $\mathcal{D}_{\text{succ}}$ separately. We also write the derivations using implicit context representations to keep them compact.

$$\begin{aligned} \mathcal{D}_z &= \frac{\frac{}{\forall x:\text{nat}. z \leq x} \text{le_z} \quad \frac{}{z : \text{nat}} \text{nat}_z}{z \leq z} \forall \text{E} \\ \mathcal{D}_{\text{succ}} &= \frac{\frac{\frac{}{\forall x:\text{nat}. \forall y:\text{nat}. x \leq y \supset \text{succ } x \leq \text{succ } y} \text{le_suc} \quad \frac{}{n : \text{nat}}}{\forall y:\text{nat}. n \leq y \supset \text{succ } n \leq \text{succ } y} \forall \text{E} \quad \frac{}{n : \text{nat}} \forall \text{E} \quad \frac{}{n \leq n} \text{ih}}{n \leq n \supset \text{succ } n \leq \text{succ } n} \supset \text{E} \quad \frac{}{\text{succ } n \leq \text{succ } n} \end{aligned}$$

6.1.3 Proof terms

We assign a recursive program to the induction rule.

$$\frac{\Gamma \vdash t : \text{nat} \quad \Gamma \vdash M_z : A(z) \quad \Gamma, n:\text{nat}, f n:A(n) \text{ true} \vdash M_{\text{succ}} : A(\text{succ } n)}{\Gamma \vdash \text{rec}^{\forall x:\text{nat}. A(x)} t \text{ with } f \ z \rightarrow M_z \mid f \ (\text{succ } n) \rightarrow M_{\text{succ}} : A(t)} \text{natE}^{n,fn}$$

The proof term uses the variable f to denote the function we are defining; in some sense, this definition is similar to programs allowing defining functions by equations using simultaneous pattern as in Haskell.

From the proof above for $\mathcal{S} \vdash \forall x:\text{nat}. x \leq x$ we can synthesize the following program:

$$\begin{array}{l} \lambda a:\text{nat}. \text{rec}^{\forall x:\text{nat}. x \leq x} a \text{ with} \\ \quad | f z \quad \Rightarrow \quad \text{le_z } z \\ \quad | f (\text{suc } n) \Rightarrow \quad \text{le_suc } n \, n \, (f \, n) \end{array}$$

Proving Predecessor We will next prove next $\forall x:\text{nat}. \neg(x = z) \supset \exists y : \text{nat}. \text{suc } y = x$. For simplicity, we define equality using reflexivity: $\text{ref} : \forall x:\text{nat}. x = x$.

\mathcal{D}

$$\frac{\mathcal{S}, a : \text{nat} \vdash \neg(a = z) \supset \exists y : \text{nat}. \text{suc } y = a}{\mathcal{S} \vdash \forall x:\text{nat}. \neg(x = z) \supset \exists y : \text{nat}. \text{suc } y = x} \forall I^a$$

To prove \mathcal{D} we will use the rule $\text{natE}^{n, f n}$; in particular, we need to prove the following base and step case:

Base case:

$$\frac{\frac{\frac{}{\neg(z = z)} \text{u} \quad \frac{}{z = z} \text{ref}}{\supset E} \quad \perp}{\frac{\exists y:\text{nat}. \text{suc } y = z}{\supset I^u} \perp E} \supset E$$

Step case:

$$\frac{\frac{\frac{}{\text{suc}(n) = \text{suc}(n)} \text{ref} \quad \frac{}{n : \text{nat}}}{\exists I} \exists y:\text{nat}. \text{suc}(y) = \text{suc}(n)}{\neg(\text{suc}(n) = z) \supset \exists y:\text{nat}. \text{suc}(y) = \text{suc}(n)} \supset I^u$$

Therefore, we have established $\forall x:\text{nat}. \neg(x = z) \supset \exists y : \text{nat}. \text{suc } y = x$. Note that we only used the induction rule to split the proof into two different cases, but we did not actually use the induction hypothesis in the proof.

It is particularly interesting to extract the corresponding proof term; we omit the type annotation on the rec for better readability.

$$\begin{array}{l} \lambda a:\text{nat. rec } a \text{ with} \\ | f z \quad \Rightarrow \lambda u:\neg(x = z).\text{abort } (u \text{ (refl } z)) \\ | f (\text{suc } n) \Rightarrow \lambda u:\neg(x = z).\langle n, \text{refl } z \rangle \end{array}$$

Can you guess what program it implements? - If we erase all subterms pertaining propositions, it becomes even more obvious:

$$\begin{array}{l} \lambda a:\text{nat. rec } a \text{ with} \\ | f z \quad \Rightarrow \text{--} \\ | f (\text{suc } n) \Rightarrow n \end{array}$$

We have obtained a verified predecessor function!

How to extend our notion of computation? We will have two reduction rules which allow us to reduce a recursive program:

$$\begin{array}{l} \text{rec}^A z \quad \text{with } f z \rightarrow M_z \mid f (\text{suc } n) \rightarrow M_{\text{suc}} \quad \Longrightarrow \quad M_z \\ \text{rec}^A (\text{suc } t) \quad \text{with } f z \rightarrow M_z \mid f (\text{suc } n) \rightarrow M_{\text{suc}} \quad \Longrightarrow \quad [t/n][r/f n]M_{\text{suc}} \\ \text{where } r = \text{rec}^A t \text{ with } f z \rightarrow M_z \mid f (\text{suc } n) \rightarrow M_{\text{suc}} \end{array}$$

Note that we unroll the recursion by replacing the reference to the recursive call $f n$ with the actual recursive program $\text{rec}^A t$ with $f z \rightarrow M_z \mid f (\text{suc } n) \rightarrow M_{\text{suc}}$.

We might ask, how to extend our congruence rules. In our setting, where reductions can happen at any given sub-term, we will have two additional congruence rules. Note that we do not have a rule which evaluates t , the term we are recursing over; at the moment, the only possible terms we can have are those formed by z and suc or variables. We have no computational power on for terms t of our domain.

Congruence rules

$$\begin{array}{c} \frac{N_z \Longrightarrow N'_z}{\text{rec}^A t \text{ with } f z \rightarrow N_z \mid f (\text{suc } n) \rightarrow N_{\text{suc}} \Longrightarrow \text{rec}^A t \text{ with } f z \rightarrow N'_z \mid f (\text{suc } n) \rightarrow N_{\text{suc}}} \\ \frac{N_{\text{suc}} \Longrightarrow N'_{\text{suc}}}{\text{rec}^A t \text{ with } f z \rightarrow N_z \mid f (\text{suc } n) \rightarrow N_{\text{suc}} \Longrightarrow \text{rec}^A t \text{ with } f z \rightarrow N_z \mid f (\text{suc } n) \rightarrow N'_{\text{suc}}} \end{array}$$

Proving subject reduction We also revisit subject reduction, showing that the additional rules for recursion preserve types. This is a good check that we didn't screw up.

Theorem 6.1.1. *If $M \Rightarrow M'$ and $\Gamma \vdash M : C$ then $\Gamma \vdash M' : C$.*

Proof. By structural induction on $M \Rightarrow M'$.

Case $\mathcal{D} = \text{rec}^A (\text{suc } t) \text{ with } f \ z \rightarrow M_z \mid f (\text{suc } n) \rightarrow M_{\text{suc}} \Rightarrow [t/n][r/f n]M_{\text{suc}}$
 where $r = \text{rec}^A t \text{ with } f \ z \rightarrow M_z \mid f (\text{suc } n) \rightarrow M_{\text{suc}}$

$\Gamma \vdash \text{rec}^A (\text{suc } t) \text{ with } f \ z \rightarrow M_z \mid f (\text{suc } n) \rightarrow M_{\text{suc}} : C$ by assumption
 $\Gamma \vdash \text{suc } t : \text{nat}$
 $\Gamma \vdash M_z : A(z)$
 $\Gamma, n:\text{nat}, f n : A(n) \vdash M_{\text{suc}} : A(\text{suc } n)$ where $C = A(\text{suc } t)$ by inversion on natE
 $\Gamma \vdash t : \text{nat}$ by natI_{suc}
 $\Gamma \vdash \text{rec}^A t \text{ with } f \ z \rightarrow M_z \mid f (\text{suc } n) \rightarrow M_{\text{suc}} : A(t)$ by rule natE
 $[t/n][r/f n]M_{\text{suc}} : A(\text{suc } t)$ by substitution lemma (twice)

\mathcal{D}'

Case $\mathcal{D} = \frac{N_z \Rightarrow N'_z}{\text{rec}^A t \text{ with } f \ z \rightarrow N_z \mid f (\text{suc } n) \rightarrow N_{\text{suc}} \Rightarrow \text{rec}^A t \text{ with } f \ z \rightarrow N'_z \mid f (\text{suc } n) \rightarrow N_{\text{suc}}}$

$\Gamma \vdash \text{rec}^A t \text{ with } f \ z \rightarrow N_z \mid f (\text{suc } n) \rightarrow N_{\text{suc}} : C$ by assumption
 $\Gamma \vdash t : \text{nat}$
 $\Gamma, n:\text{nat}, f n : A(n) \vdash N_{\text{suc}} : A(\text{suc } n)$
 $\Gamma \vdash N_z : A(z)$ where $C = A(t)$ by inversion on natE
 $\Gamma \vdash N'_z : A(z)$ by i.h.
 $\Gamma \vdash \text{rec}^A t \text{ with } f \ z \rightarrow N'_z \mid f (\text{suc } n) \rightarrow N_{\text{suc}} : C$ by rule natE \square

6.1.4 Programming Proofs

First-order logic gives rise to dependent types and inductive proofs correspond to recursive programs. How can then program these inductive proofs? - We revisit this question using the programming and proof environment Beluga.

Beluga is a two-level system: on the first level, we encode domain specific knowledge such as the formula defining \leq ; on the level above, we reason about such specification using first-order logic and recursion.

Let us look at the previous example where we proved that all natural numbers are less than or equal to themselves, i.e.

For all $x:\text{nat}$. $x \leq x$.

We know how to encode \leq as a type family in LF as follows:

```
LF leq: nat → nat → type =
| le_z: leq z N
| le_s: leq N M → leq (s N) (s M);
```

The theorem is encoded as the type of a recursive function about the domain \mathcal{S} that contains the definitions about nat and leq.

```
rec ref: {N: [⊢ nat]} [⊢ leq N N] = ? ;
```

Universal quantification is written using curly braces in Beluga. We further wrap any domain definitions we are using in brackets [].

In the proof, we first introduced N. The proof term for introducing a universal quantifier in Beluga is **mlam** $N \Rightarrow \dots$. We then split on N and distinguish the case where N is z and where N is s M.

If $N = z$, then we must return a witness for $\vdash \text{leq } z \ z$ which is `le_z`.

If $N = s \ M$, then by recursion (induction) we obtain a proof P for $\vdash \text{leq } M \ M$. We need to return a proof for $\vdash \text{leq } (s \ M) \ (s \ M)$ which is described by $\vdash \text{le_s } P$.

The full Beluga program implementing the proof as a recursive program is written below:

```
rec ref: {N: [⊢ nat]} [⊢ leq N N] =
/ total n (ref n) /
mlam N  $\Rightarrow$  case [⊢ N] of
| [⊢ z]  $\Rightarrow$  [⊢ le_z]
| [⊢ s M]  $\Rightarrow$  let [⊢ P] = ref [⊢ M] in [⊢ le_s P];
```

We further annotate the function with `/ total n (ref n) /` which asks Beluga to check that the function `ref` is a total function which recurses over the first argument. To verify the function is total, Beluga makes sure we have covered all cases and all recursive calls are on a number that is smaller than N.

Remark 1 All universally quantified variables such as N must be capitalized. This is a syntactic convention in Beluga.

Remark 2 Any derivation we describe inside [] can only be built by either constants from our domain or variables that stand for domain objects. We cannot write `[⊢ le_s (ref [⊢ M])]`.